



Calculate Child's Age in Months at the End of the Investigation

NSCAW Restricted Release dataset, CPS sample

Based on SPSS version 15.0

The NSCAW Restricted Release dataset provides the child's age in months at the time of each interview, but because of confidentiality concerns, it does not include the child's date of birth. Researchers are often interested in knowing how old the NSCAW children were at other times. The following bit of SPSS syntax illustrates how a user can compute children's age in months at the end of the CPS investigation that brought them into the NSCAW, an event that takes place a two or more months before the Wave 1 interview. After creating date-formatted variables, the program computes the number of months between the end of the investigation and the Wave 1 interview. It then subtracts this interval from children's age at Wave 1 to arrive at age at the end of the investigation.

```
***** Using SPSS *****.
*** Convert string dates to date variables.
do repeat dattxt= invdate ycompdat pcompdat /datevar=invEnd ycomp pcomp.
    compute datevar = date.mdy(
        number(substr(dattxt,1,2),f2),
        number(substr(dattxt,3,2),f2),
        number(substr(dattxt,5,4),f4)
    ).
    formats datevar (adate10).
end repeat.
freq invEnd ycomp pcomp.

*** Compute interval (in months) between investigation end date and wave 1 completion
date.
* Age in the child instrument is used if it's not missing; if it is, age in the
caregiver instrument is used.
compute interval=-9.
do if (missing(ycomp)=0).
    compute interval=datediff(ycomp,invEnd, "months").
else if (missing(ycomp)=1).
    compute interval=datediff(pcomp,invEnd, "months").
end if.
* Negative, illegal intervals are treated as missing values.
missing values interval (lo thru -1).
freq interval.

*** Subtract interval from age in months to get age at investigation end.
* Age will not be computed if any intervals are negative.
compute invEndAge=chdagem-interval.
formats invEndAge (f8.0).
variable labels invEndAge "Age at End of Investigation".
freq invEndAge.
```